// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (access/Ownable.sol)

pragma solidity ^0.8.20;

import {Context} from "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * The initial owner is set to the address provided by the deployer. This can
 * later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    /**
     * @dev The caller account is not authorized to perform an operation.
     */
    error OwnableUnauthorizedAccount(address account);

    /**
     * @dev The owner is not a valid owner account. (eg. `address(0)`)
     */
    error OwnableInvalidOwner(address owner);

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
     */
    constructor(address initialOwner) {
        if (initialOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(initialOwner);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        if (owner() != _msgSender()) {
            revert OwnableUnauthorizedAccount(_msgSender());
        }
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (token/ERC721/extensions/IERC721Enumerable.sol)

pragma solidity ^0.8.20;

import {IERC721} from "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {
    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.20;

import {IERC165} from "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon
     *   a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or
     *   {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon
     *   a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Note that the caller is responsible to confirm that the recipient is capable of receiving ERC721
     * or else they may be permanently lost. Usage of {safeTransferFrom} prevents loss, though the caller must
     * understand this adds an external call which potentially creates a reentrancy vulnerability.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the address zero.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/Context.sol)

pragma solidity ^0.8.20;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/introspection/IERC165.sol)

pragma solidity ^0.8.20;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: contracts/interface/IWETH.sol



pragma solidity >=0.5.0;

interface IWETH {
    function deposit() external payable;
    function transfer(address to, uint value) external returns (bool);
    function transferFrom(address from, address to, uint value) external returns (bool);
    function withdraw(uint) external;
}
// File: contracts/Treasury.sol



pragma solidity 0.8.20;

import "./interface/IWETH.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol";

/// @title   IntrinsicTreasury
/// @notice  Treasury for Intrinsic collection
contract IntrinsicTreasury is Ownable {
    /// EVENTS ///

    /// @notice        Emitted after RFV has been updated
    /// @param newRFV  New RFV of collection
    event RFVChanged(uint256 indexed newRFV);

    /// @notice                 Emitted after backing has been added
    /// @param backingAdded     Amount of backing added
    /// @param newTotalBacking  New total backing
    event BackingAdded(uint256 backingAdded, uint256 newTotalBacking);

    /// @notice                 Emitted after item has been redeemed
    /// @param itemId           Id of item redeemed
    /// @param newTotalBacking  New total backing
    event ItemRedeemed(uint256 itemId, uint256 newTotalBacking);

    /// @notice                 Emitted after item has been purchased
    /// @param itemId           Id of item purchased
    /// @param newTotalBacking  New total backing
    event ItemPurchased(uint256 itemId, uint256 newTotalBacking);

    /// ERRORS ///
    /// @notice Error for if invalid input
    error InvalidInput();
    /// @notice Error for if address is already set
    error AddressAlreadySet();
    /// @notice Error for if NFT is not treasury owned
    error NotTreasuryOwned();
    /// @notice Error for if not owner
    error NotOwner();
    /// @notice Error for if no float value is 0
    error NoFloatValueZero();

    /// STATE VARIABLES ///

    /// @notice Address of WETH
    address public immutable WETH;
    /// @notice Address of collection
    IERC721Enumerable public collection;

    /// @notice Fee percent for purchasing through treasury
    uint256 public immutable royaltyPercent;

    /// @notice Current backing of the treasury
    uint256 public backing;
    /// @notice Value to sell first NFT if float is 0
    uint256 public noFloatValue;
    /// @notice Number of items treasury owns
    uint256 public itemsTreasuryOwns;

    /// @notice Bool if item id is treasury owned
    mapping(uint256 => bool) public treasuryOwned;

    /// CONSTRUCTOR ///

    /// @param weth_            Address of weth
    /// @param royaltyPercent_  Royalty percent
    constructor(address weth_, uint256 royaltyPercent_) Ownable(msg.sender) {
        WETH = weth_;
        royaltyPercent = royaltyPercent_;
    }

    /// @notice             Function that allows owner to set collection of treasury
    /// @param collection_  Address of collection for treasury
    function setCollection(address collection_) external onlyOwner {
        if (address(collection) != address(0)) revert AddressAlreadySet();

        collection = IERC721Enumerable(collection_);
    }

    /// EXTERNAL FUNCTIONS ///

    /// @notice         Function that adds wETH to backing
    /// @param amount_  Amont of wETH to add to backing
    function addToBacking(uint256 amount_) external {
        if (amount_ == 0) revert InvalidInput();
        IWETH(WETH).transferFrom(msg.sender, address(this), amount_);
        backing += amount_;

        emit BackingAdded(amount_, backing);
        emit RFVChanged(realFloorValue());
    }

    /// @notice       Function that redeems item to treasury to receive backing
    /// @param id_    Item id to redeem
    function redeemItem(uint256 id_) external {
        if (treasuryOwned[id_]) revert NotOwner();

        uint256 rfv_ = realFloorValue();

        backing -= rfv_;

        ++itemsTreasuryOwns;
        treasuryOwned[id_] = true;

        collection.transferFrom(msg.sender, address(this), id_);
        IWETH(WETH).transfer(msg.sender, rfv_);

        emit ItemRedeemed(id_, backing);
        emit RFVChanged(realFloorValue());
    }

    /// @notice       Function that purchases item from treasury
    /// @param id_    Item id to purchase
    function purchaseItem(uint256 id_) external {
        if (float() == 0 && noFloatValue == 0) revert NoFloatValueZero();
        if (!treasuryOwned[id_]) revert NotTreasuryOwned();

        uint256 rfv_ = realFloorValue();
        uint256 fee_ = (royaltyPercent * rfv_) / 100;
        uint256 toPay_ = rfv_ + fee_;
        uint256 toBacking_ = toPay_;

        backing += toBacking_;

        --itemsTreasuryOwns;
        treasuryOwned[id_] = false;

        IWETH(WETH).transferFrom(msg.sender, address(this), toPay_);
        collection.transferFrom(address(this), msg.sender, id_);

        emit ItemPurchased(id_, backing);
        emit RFVChanged(realFloorValue());
    }

    /// EXTERNAL VIEW FUNCTIONS ///

    /// @notice         Function that returns RFV of items in collection
    /// @return value_  RFV of item in collection
    function realFloorValue() public view returns (uint256 value_) {
        uint256 backing_ = backing;
        uint256 float_ = float();

        if (float_ > 0) value_ = backing_ / float_;
        else value_ = backing_ + noFloatValue;
    }

    /// @notice         Function that returns float of collection
    /// @return float_  Number of supply not treasury owned
    function float() public view returns (uint256 float_) {
        float_ = collection.totalSupply() - itemsTreasuryOwns;
    }

    /// OWNER FUNCTIONS ///

    /// @notice  Function that sets purchase price if float is 0
    function setNoFloatValue(uint256 value_) external onlyOwner {
        if (value_ == 0) revert NoFloatValueZero();
        noFloatValue = value_;

        emit RFVChanged(realFloorValue());
    }
}